



Open Data Description Language


openddl.org




```
structure ::= data-type (name? ("{" data-list? "}") | "[" integer-literal "]" "*" name? ("{" data-array-list? "}") | identifier name? ("(" (property ("," property)*?) ")")? ("{" structure* "}")
```




```
data-type ::=
"bool" | "b" |
"int8" | "i8" |
"int16" | "i16" |
"int32" | "i32" |
"int64" | "i64" |
"uint8" | "u8" |
"uint16" | "u16" |
"uint32" | "u32" |
"uint64" | "u64" |
"half" | "h" |
"float16" | "f16" |
"float" | "f" |
"float32" | "f32" |
"double" | "d" |
"float64" | "f64" |
"string" | "s" |
"ref" | "r" |
"type" | "t" |
"base64" | "z"
```




```
identifier ::= [A-Za-z_] [A-Za-z0-9]*
```




```
name ::= ("$" | "%") identifier
```




```
reference ::= name ("% identifier)* | "null"
```




```
base64-data ::= ([A-Za-z0-9] | "+" | "/" | "=")*
```




```
integer-literal ::= ("+" | "-")?
(decimal-literal | hex-literal | octal-literal |
binary-literal | char-literal)
```




```
escape-char ::= '\ ' | '\ ' | '\ ' | '\ ' | '\ ' |
'\a' | '\b' | '\f' | '\n' | '\r' | '\t' | '\v' |
'\x' hex-digit hex-digit
```




```
decimal-literal ::=
[0-9] ("_"? [0-9])*
```




```
hex-literal ::=
("0x" | "0X")
hex-digit
("_" hex-digit)*
```




```
octal-literal ::=
("0o" | "0O")
[0-7] ("_"? [0-7])*
```




```
binary-literal ::=
("0b" | "0B")
("0" | "1")
("_" ("0" | "1"))*
```




```
char-literal ::=
"'" ([#x20-#x26#x28-#x5B#x5D-#x7E] | escape-char)+ "'"
```




```
data-list ::=
bool-literal ("," bool-literal)* |
integer-literal ("," integer-literal)* |
float-literal ("," float-literal)* |
string-literal ("," string-literal)* |
reference ("," reference)* |
data-type ("," data-type)* |
base64-data ("," base64-data)*
```




```
bool-literal ::=
"false" | "0" |
"true" | "1"
```




```
property ::=
identifier "="
bool-literal |
integer-literal |
float-literal |
string-literal |
reference |
data-type |
base64-data
```



```
hex-digit ::= [0-9A-Fa-f]
```



```
float-literal ::= ("+" | "-")? ([0-9] ("_"? [0-9])* ("." [0-9] ("_"? [0-9])*)?)? | "." [0-9] ("_"? [0-9])*
(("e" | "E") ("+" | "-")? [0-9] ("_"? [0-9])*)? | hex-literal | octal-literal | binary-literal
```



```
string-literal ::= ('"' ([#x20-#x21#x23-#x5B#x5D-#x7E#xA0-#xD7FF#xE000-#xFFFF#x10000-#x10FFFF] | escape-char |
"\u" hex-digit hex-digit hex-digit hex-digit | "\U" hex-digit hex-digit hex-digit hex-digit hex-digit hex-digit)+ '"')
```